
1 Faster Division

Quotient and Remainder

• Recall the problem:

con A,B : Int {0 ⩽ A ∧ 0< B}
var q , r : Int
?
{A = q × B + r ∧ 0 ⩽ r < B} .

• Recall: recognising the postcondition as a conjunc-
tion, we useA = q×B+r ∧ 0 ⩽ r as the invariant
and ¬ (r < B) as the guard.

• The program we came up with:

q , r := 0,A
{A = q × B + r ∧ 0 ⩽ r , bnd : r}
do B ⩽ r → q := q + 1

r := r − B
od
{A = q × B + r ∧ 0 ⩽ r < B} .

• In each iteration of the loop, r is decreased by B .

• We can probably get a quicker program by decreas-
ing r by ... 2× B , when possible.

• What about decreasing r by 4× B , 8× B ,... etc?

1.1 Division in O(log(A/B)) Time

Strategy...

con A,B : Int {0 ⩽ A ∧ 0< B}
var q , r , b, k : Int
...
{0 ⩽ k ∧ b = 2k × B ∧ A< b}
...
{A = q × b + r ∧ 0 ⩽ r < b ∧
0 ⩽ k ∧ b = 2k × B , bnd : b}

do b ̸= B → ...od
{A = q × B + r ∧ 0 ⩽ r < B}

Generating 2k × B

• It is easy to satisfy b = 2k × B ∧ A< b.

b, k := B , 0
do b ⩽ A → b, k := b × 2, k + 1 od
{0 ⩽ k ∧ b = 2k × B ∧ A< b}

• What are the loop invariant and the bound?

• Initialisation for the next loop easily follows:

{0 ⩽ k ∧ b = 2k × B ∧ A< b}
q , r := 0,A
{A = q × b + r ∧ 0 ⩽ r < b ∧
0 ⩽ k ∧ b = 2k × B}

Decreasing b

• What needs to be done before we decrement b by
half?

(A = q × b + r ∧ 0 ⩽ r < b)[b\b / 2]
≡ (A = q × (b / 2) + r ∧ 0 ⩽ r < b / 2)

• We can restore the invariant by q := q × 2...

(A = q × (b / 2) + r ∧ 0 ⩽ r < b / 2)[q\q × 2]
≡ A = (q × 2)× (b / 2) + r ∧ 0 ⩽ r < b / 2
⇐ A = q × b + r ∧ 0 ⩽ r < b / 2 ∧ b = 2k × B

• only if we already have r < b / 2!

• That gives us one guarded command:

r < b / 2 → q , b, k := q × 2, b / 2, k − 1

Decreasing b — The Other Case

• What about the case when b / 2 ⩽ r < b?

• The task is to find a substitution such that

(A = q × (b / 2) + r ∧ 0 ⩽ r < b / 2)[?\?]
⇐ A = q × b + r ∧ b / 2 ⩽ r < b ∧ b = 2k × B

• Comparing 0 ⩽ r<b/2 and b/2 ⩽ r<b, onemight
want to try a substitution containing [r\r − b / 2].

(0 ⩽ r < b / 2)[r\r − b / 2]
≡ 0 ⩽ r − b / 2< b / 2
⇐ b / 2 ⩽ r < b ∧ b = 2k × B .

• Consider the former half of the expression:

(A = q × (b / 2) + r)[r\r − b / 2]
≡ A = q × (b / 2) + r − b / 2
≡ A = (q − 1)× (b / 2) + r .

• Applying [q\q×2+1] gives us backA = q×b+ r .

• Therefore, another guarded command:

b / 2 ⩽ r → q , b, k , r :=
q × 2 + 1, b / 2, k − 1, r − b / 2

The Program

con A,B : Int {0 ⩽ A ∧ 0< B}
var q , r , b, k : Int

b, k := B , 0
do b ⩽ A → b, k := b × 2, k + 1 od
{0 ⩽ k ∧ b = 2k × B ∧ A< b}
q , r := 0,A
{A = q × b + r ∧ 0 ⩽ r < b ∧
0 ⩽ k ∧ b = 2k × B , bnd : b}

do b ̸= B →
if r < b / 2 → q , b, k := q × 2, b / 2, k − 1
| b / 2 ⩽ r → q , b, k , r := q × 2 + 1, b / 2,

k − 1, r − b / 2
fi

od
{A = q × B + r ∧ 0 ⩽ r < B}

1.2 Alternative Programs
ExistentialQuantification

• The variable k is used in the proofs, but not needed
for computing the output.

• Such a variable is called a “ghost variable” in Kalde-
waij [Kal90].

• One can remove k , and the program would still
work.

• For its reasoning, we need to use existential quan-
tification in the assertions to talk about properties
involving k .

con A,B : Int {0 ⩽ A ∧ 0< B}
var q , r , b : Int

b := B
do b ⩽ A → b := b × 2 od
{⟨∃k : 0 ⩽ k : b = 2k × B⟩ ∧ A< b}
q , r := 0,A
{A = q × b + r ∧ 0 ⩽ r < b ∧
⟨∃k : 0 ⩽ k : b = 2k × B⟩, bnd : b}

do b ̸= B →
if r < b / 2 → q , b := q × 2, b / 2
| b / 2 ⩽ r → q , b, r := q × 2 + 1, b / 2,

r − b / 2
fi

od
{A = q × B + r ∧ 0 ⩽ r < B}

In developing such programs,

• We can introduce variables such as k , and realise
that they are ghost variables and remove them later.

• Or we can have existential quantification in asser-
tions to begin with, if you are sure that the quanti-
fied variables won’t be needed.

Alternative Program
Kaldewaij [Kal90] presented the following alternative.

Do you prefer this program?

con A,B : Int {0 ⩽ A ∧ 0< B}
var q , r , b, k : Int

b, k := B , 0
do b ⩽ A → b, k := b × 2, k + 1 od
q , r := 0,A
do b ̸= B →
q , b, k := q × 2, b / 2, k − 1
if r < b → skip
| b ⩽ r → q , r := q + 1, r − b
fi

od
{A = q × B + r ∧ 0 ⩽ r < B}

• The program has the advantage that we do not need
to have b / 2 in the guards.

• Note what the first assignment establishes:

{A = q × b + r ∧ 0 ⩽ r < b ∧
0 ⩽ k ∧ b = 2k × B ∧ b ̸= B}

q , b, k := q × 2, b / 2, k − 1
{A = q × b + r ∧ 0 ⩽ r < 2× b ∧

0 ⩽ k ∧ b = 2k × B}

A Historical Note

• The correctness of the if in the loop was actually a
key example in Dahl [DDH72], one of the earliest
book on structured programming:

{0 ⩽ r < b}
b := b / 2
if r < b → skip
| b ⩽ r → r := r − b
fi
{0 ⩽ r < b}

• In Dahl [DDH72], Dijkstra needed about one page
of textual proof.

• These days we can prove its correctness by routine
symbolic manipulation. It shows how much sym-
bolic reasoning has advanced since then.

2 Binary Search Revisited

Binary Search

• Given a sorted array ofN numbers and a key, either
locate the position where the key resides in the ar-
ray, or report that the key does not present in the
array, in O(logN) time.

• A possible spec:

con N ,K : Int {0<N }
con F : array [0..N ) of Int {F ascending}
var l , r : Int
bsearch
{F [l ] = K ∨ ...} .

2.1 The van Gasteren-Feijen Approach

• Van Gasteren and Feijen [vGF95] pointed a surpris-
ing fact: binary search does not apply only to sorted
lists!

• In fact, they believe that comparing binary search
to searching for a word in a dictionary is a major
educational blunder.

• Their binary search: let Φ be a predicate on two in-
tegers with some additional constraints to be given
later:

con M ,N : Int {M <N ∧ Φ M N ∧ ...}
var l , r : Int
bsearch
{M ⩽ l <N ∧ Φ l (l + 1)} .

Invariant and Bound

• Invariant: Φ l r ∧ M ⩽ l < r ⩽ N , loop guard:
l + 1 ̸= r .

• Initialisation: l , r :=M ,N .

• Bound: r − l .

• For anym such that l<m<r , we have r−m<r−l
andm− l<r− l . Therefore both l :=m and r :=m
decrease the bound.

Constructing the Loop Body

• For l :=m we calculate.

(Φ l r ∧ M ⩽ l < r ⩽ N )[l\m]
≡ Φ m r ∧ M ⩽ m < r ⩽ N
⇐ Φ m r ∧ M ⩽ l <m < r ⩽ N .

• That l<m<r is our assumption. The leftoverΦ m r
gives rise to a guarded command: Φ m r → l :=m .

• The case with r :=m is similar.

The Program Skeleton

{M <N ∧ Φ M N }
l , r :=M ,N
{Φ l r ∧ M ⩽ l < r ⩽ N , bnd : r − l}
do l + 1 ̸= r →
{... ∧ l + 2 ⩽ r}
m := anything s.t. l < m < r
{... ∧ l <m < r}
if Φ m r → l :=m
| Φ l m → r :=m
fi

od
{M ⩽ l <N ∧ Φ l (l + 1)}

Note: m := (l + r) / 2 is a valid choice, thanks to the
precondition that l + 2 ⩽ r .

Constraints on Φ

• But we need the if to be total.

• Therefore we demand a constrant on Φ:

Φ l r ⇒ Φ l m ∨ Φ m r , if l < m < r. (1)

• Some Φ satisfying (1) (for F of appropriate type):

– Φ l r ≡ F [l ] ̸= F [r ],

– Φ l r ≡ F [l ]< F [r ],

– Φ l r ≡ F [l ] ⩽ A ∧ A ⩽ F [r ],

– Φ l r ≡ F [l ]× F [r ] ⩽ 0,

– Φ l r ≡ F [l ] ∨ F [r ],

– Φ l r ≡ ¬ (Q l) ∧ Q r .

• Van Gasteren and Feijen believe that Φ l r =
F [l ] ̸= F [r ] is a better example when explaining
binary search.

2.2 Searching for a Key

• The case Φ l r ≡ ¬ (Q l) ∧ Q r is worth special
attention.

• Choose Q i ≡ K < F [i ] for some K .

• Therefore Φ l r ≡ F [l ] ⩽ K < F [r ].

• That constitutes the binary search we wanted!

• The postcondition: M ⩽ l<N ∧ F [l ] ⩽ K <F [l+
1].

• Note that we do not yet need F to be sorted!

• The algorithm gives you some l such that F [l ] ⩽
K <F [l +1]. If there are more than one such l , one
is returned non-deterministically.

Sortedness

• ThatF is sorted comes in whenwe need to establish
that there is at most one l satisfying the postcondi-
tion.

• That is, either F [l ] = K , or ¬ ⟨∃i : M ⩽ i < N :
F [i ] = K ⟩.

The Program... Or A Part Of It

• Let Φ l r = F [l ] ⩽ K < F [r ].

• Processing the array fragment F [a . . b ]:

l , r := a, b
{Φ l r ∧ a ⩽ l < r ⩽ b, bnd : r − l}
do l + 1 ̸= r →
m := (l + r) / 2
if F [m] ⩽ K → l :=m
| K < F [m] → r :=m
fi

od
{a ⩽ l < b ∧ F [l ] ⩽ K < F [l + 1]}

• Note that F [a] and F [b] are never accessed.

• This program is not yet complete....

Initialisation

• But wait.. to apply the algorithm to the entire array,
we need the precondition Φ 0 N , that is F [0] ⩽
K <F [N ]. Is that true? (We don’t even have F [N ].)

• One can rule out cases when the precondition do
not hold (and also deal with empty array). E.g.

if 0 = N → p := False
| 0<N →
if K < F [0] → p := False
| F [N − 1] = K → p, l := True,N − 1
| F [0] ⩽ K < F [N − 1] →

a, b := 0,N − 1
program above
p := F [l ] = K

fi
fi

• where p is True iff. K presents in F .

Pseudo Elements

• But there is a better way... introduce two pseudo
elements!

• Let F [−1] = −∞ and F [N ] = ∞.

• Initially, Φ (−1) N is satisfied.

• In the code, F [−1] and F [N ] are never accessed.
Therefore we do not actually have to represent
them!

• We need to be careful interpreting the result, once
the main loop terminates, however.

The Program (1)
Let Φ l r = F [l ] ⩽ K < F [r ].

con N ,K : Int {0 ⩽ N }
con F : array [0..N ) of Int {F ascending ∧
F [−1] = −∞ ∧ F [N ] = ∞}

var l ,m, r : Int
var p : Bool

l , r :=−1,N
{Φ l r ∧ −1 ⩽ l < r ⩽ N , bnd : r − l}
do l + 1 ̸= r →
m := (l + r) / 2
if F [m] ⩽ K → l :=m
| K < F [m] → r :=m
fi

od
{−1 ⩽ l <N ∧ F [l ] ⩽ K < F [l + 1]}

The Program (2)

{−1 ⩽ l <N ∧ F [l ] ⩽ K < F [l + 1]}
if − 1 = l → p := False
| 0 ⩽ l → p := F [l ] = K
fi
{p = ⟨∃i : 0 ⩽ i <N : F [i ] = K ⟩ ∧
p ⇒ F [l ] = K}

Alternative Program

• Kaldewaij [Kal90, Sec. 6.3] derived an alternative
program that introduces only F [N ] = ∞ (but not
F [−1] = −∞), while requiring the array to be non-
empty.

• The main loop is the same. It is only post-loop in-
terpretation that is different.

2.3 Searching with Premature Return

A More Common Program

• Recall that Bentley [Ben86, pp. 35-36] proposed us-
ing binary search as an exercise.

• Bentley’s solution can be rephrased below:

l , r , p := 0,N − 1,False
do l ⩽ r →
m := (l + r) / 2
if F [m]<K → l :=m + 1
| F [m] = k → p := True; break
| K < F [m] → r :=m − 1
fi

od

A More Common Program
I’d like to derive it, but

• it is harder to formally deal with break .

– Still, Bentley employed a semi-formal reason-
ing using a loop invariant to argue for the cor-
rectness of the program.

• To relate the test F [m] < K to l := m+ 1 we have
to bring in the fact that F is sorted earlier.

Comparison

• The two programs do not solve exactly the same
problem (e.g. when there are multipleKs in F ).

• Is the second program quicker because it assigns l
and r tom+ 1 and m− 1 rather thanm?

– l := m+1 because F [m] is covered in another
case;

– r := m − 1 because a range is represented
differently.

• Is it quicker to perform an extra test to return
early?

– When K is not in F , the test is wasted.

– Rolfe [Rol97] claimed that single comparison
is quicker in average.

– Knuth [Knu97, Exercise 23, Section 6.2.1]: sin-
gle comparison needs 17.5 lgN + 17 instruc-
tions, double comparison needs 18 lgN − 16
instructions.

